
ABSTRACT 
Cumulative incidence ratio (CIR) is a measure of association that is the ratio of two cumulative incidences, or the 
proportion of one group (e.g. a treated group) who experiences an event relative to the proportion of another group 
(e.g. a control group) who experiences the event. This paper shows how to derive CIR and a point-wise 95% 
confidence interval with the aid of SAS® procedure PROC LIFETEST, and how to use the Miettinen-Nurminen 
solution to handle some special cases. A graphic representation that aims to better detect the group difference is 
introduced.  

INTRODUCTION 
It is difficult to extract useful information from a crowded Kaplan-Meier plots (e.g. multiple curves and/or with 
confidence intervals for each group). In particular, overlapping confidence intervals of Kaplan-Meier curves does not 
directly correspond statistically to equal curves at the 95% level. A cumulative incidence ratio plot is a good 
supplement in this case.  
 
Cumulative incidence is a measure of the probability or risk of event (e.g. disease recurrence). It indicates what 
proportion of the population will experience the event during the specified time period. The cumulative incidence ratio 
is typically the ratio of the cumulative incidence in a treated or exposed group of people to that in a control or 
unexposed group of people over the same time period. It is the same as the relative risk (Gail and Benichou 2000). 
Based on the variables provided by PROC LIFETEST, cumulative incidence ratio and the point-wise 95% confidence 
interval can be derived and then plotted. A CIR plot has the advantage of being visually appealing and hence is 
easier to interpret. Furthermore, the hypothesis of ratio=1 (e.g. treatment effect is equal to the control effect) 
corresponds quite closely to the 95% confidence interval. 

DERIVATION 

The following table shows the variables from PROC LIFETEST that are needed in the calculation of CIR and the 
point-wise 95% confidence interval.  
 

Variable Description Variable Name
Treated/Exposed Group Control/Unexposed Group

Kaplan-Meier survival estimates SURVt SURVp 
Standard error (Greenwood formula) SEt SEp 
Risk set size at current time point Nt Np 
 
Table 1. Variables from PROC LIFETEST to Calculate CIR and 95% Confidence Interval 
 
Then, at every time point, for SURVt<1 and SURVp<1,  
 

CIR=(1-SURVt)/(1-SURVp), 
 
and the corresponding 95% confidence interval is defined as  
 

[CIR*exp{-POW}, CIR*exp{POW}], 
 
where  
 



 

POW=1.96*sqrt{[SEt/(1-SURVt)]2+[SEp/(1-SURVp)]2}. 
 
For the case where SURVt=1 but SURVp<1 (CIR=0), the lower limit will be set to 0 and the upper limit will be given by 
the Miettinen-Nurminen solution (Miettinen and Nurminen 1985): 
 

(B+sqrt{B2+4*A*1.964*Np
5})*0.5/A, 

 
where 

A=(Np+Nt)*[Nt*(Np-1)*(1-SURVp)*Np]2+1.962*Nt*(Np-1)*(1-SURVp)*Np
3*[Nt+Np*(1-SURVp)] , 

B=1.962*Np
3*[1.962*Np*Nt+2*(Nt+Np)*Nt*(Np-1)*(1-SURVp)-Nt*(Np-1)]. 

 
If SURVt<1 but SURVp=1, then CIR and the upper limit will be assigned a large number (say 999) and the lower limit 
will be  
 

1/[(B+sqrt{B2+4*A*1.964*Np
5})*0.5/A], 

 
where A and B are defined as above. For ease of implementation, CIR is set to 1, and both limits are set to missing if 
SURVt=SURVp=0, and 1 if SURVt=SURVp=1. 
 
The algorithm is implemented as follows: 
 

*****************; 
**** CIR=1/1 ****; 
*****************; 
if surv_t eq 1 and surv_p eq 1 then do; 
  cir=1; 
  cir_l=1; 
  cir_u=1; 
end; 
***************************; 
**** CIR=a/b, a<1, b<1 ****; 
***************************; 
else if surv_t lt 1 and surv_p lt 1 then do; 
  cir=(1-surv_t)/(1-surv_p); 
  if surv_t eq 0 and se_t eq 0 and n_t eq 0 then cum_t=0;  
  else cum_t=(se_t/surv_t)**2; 
  if surv_p eq 0 and se_p eq 0 and n_p eq 0 then cum_p=0;  
  else cum_p=(se_p/surv_p)**2; 
  pow=1.96*sqrt(cum_t*(surv_t/(1-surv_t))**2+cum_p*(surv_p/(1-surv_p))**2); 
  cir_l=cir*exp(-pow); 
  cir_u=cir*exp(pow); 
end; 
*************************************; 
**** Miettinen-Nurminen Solution ****; 
*************************************; 
else if (surv_t eq 1 and surv_p lt 1) or (surv_t lt 1 and surv_p eq 1) then do; 
  aa=(n_p+n_t)*(n_t*(n_p-1)*(1-surv_p)*n_p)**2+(1.96**2)*n_t*(n_p-1)*(1-surv_p)* 
     (n_p**3)*(n_t+n_p*(1-surv_p));  
  bb=(1.96**2)*(n_p**3)*((1.96**2)*n_p*n_t+2*(n_t+n_p)*n_t*(n_p-1)*(1-surv_p)- 
     n_t*(n_p-1)); 
  mn_soln=(bb+sqrt(bb**2+4*aa*(1.96**4)*n_p**5))*0.5/aa; 
  **********************; 
  **** CIR=0/b, b>0 ****; 
  **********************; 
  if surv_t eq 1 and surv_p lt 1 then do; 
    cir=0; 
    cir_l=0; 
    cir_u=mn_soln; 
  end; 
  **********************; 
  **** CIR=a/0, a>0 ****; 
  **********************; 
  else if surv_t lt 1 and surv_p eq 1 then do; 



 

    cir=999; 
    cir_l=1/mn_soln; 
    cir_u=999; 
  end; 
end; 

GRAPH 
An example final graph output is in Figure 1. The event of interest in the example is the first observed significant 
difference between an active treatment group and placebo for response. The Kaplan-Meier estimates and cumulative 
incidence ratio with 95% confidence interval are integrated side-by-side for each treatment group/placebo pair for 
completeness and consistency.  
 

 
Figure 1. Cumulative Incidence Kaplan-Meier and Ratio Plot by Treatment Group 
 
Here is a streamline on how to exploit SAS/GRAPH tools to assemble the final graph. 
 

• PROC GREPLAY - Define a 3 by 2 multi-panel template and integrate all plots together 
 

proc greplay igout=work.gseg gout=work.gseg tc=mytemps nofs; 
  tdef t3x2 
  1 / llx=0  ulx=0  urx=48  lrx=48  lly=65 uly=96  ury=96  lry=65  
  2 / llx=52 ulx=52 urx=100 lrx=100 lly=65 uly=96  ury=96  lry=65  
  3 / llx=0  ulx=0  urx=48  lrx=48  lly=33 uly=63  ury=63  lry=33   
  4 / llx=52 ulx=52 urx=100 lrx=100 lly=33 uly=63  ury=63  lry=33   
  5 / llx=0  ulx=0  urx=48  lrx=48  lly=0  uly=30  ury=30  lry=0   



 

  6 / llx=52 ulx=52 urx=100 lrx=100 lly=0  uly=30  ury=30  lry=0   
  7 / llx=0  ulx=0  urx=48  lrx=48  lly=0  uly=100 ury=100 lry=0    
  8 / llx=52 ulx=52 urx=100 lrx=100 lly=0  uly=100 ury=100 lry=0    
  9 / llx=0  ulx=0  urx=100 lrx=100 lly=0  uly=100 ury=100 lry=0 
  ; 
  template t3x2; 
  tplay 1:km_0 2:cir_0 3:km_1 4:cir_1 5:km_2 6:cir_2 7:leftt 8:rightt 9:grandt; 
run; 

  
• PROC GPLOT - Generate each individual plots (panels 1 to 6) 
• PROC GSLIDE - Generate grand title/footnote (panel 9 "grandt") and panel column title/footnote (panel 7 

                             "leftt" and panel 8 "rightt") 
• Annotate Facility - Risk sets at each time point in Kaplan-Meier plots,  and , where 

 
   By calculation, the maximum value of CIR upper limit can be far away from the reference level (=1). 

When this happens, an evenly spaced y-axis ranged from minimum value of CIR lower limit to the 
maximum value of CIR upper limit usually squeezes the trend near reference level unrecognizable. 
To circumvent this, the y-axis is "bent" so that the area of interest can distinguishably stand out while 
preserving the original range and shape proportionally. 

  The time to event is defined as the assessment time at which the lower limit of the confidence interval 
of cumulative incidence ratio first goes above 1. This is dynamically marked in each CIR plot.   

 
Reference lines are also added to enhance interpretability. The vertical line in Kaplan-Meier plots indicates the end of 
study period (day 87). The horizontal line in CIR plots shows the reference level defining the study event (ratio=1).  

CONCLUSION 
The calculations of CIR and point-wise 95% confidence interval are fairly straightforward and can be readily 
implemented in SAS. Compared with the conventional Kaplan-Meier plot, the CIR plot can provide more 
comprehensive understanding of the group difference. Also, the CIR 95% confidence interval has much better 
interpretability than overlapping 95% confidence interval for Kaplan-Meier plots with respect to showing significant 
departure from the null hypothesis of no group difference. Finally, as an enhancement to the graph, summary tables 
can be appended per study need through the power of PROC GREPLAY and SAS/GRAPH Annotate Facility. 

